
Abstract

This whitepaper provides an introduction to elliptic curve cryptography for casual users while
delivering a brief tour of the Erlang language for those interested in functional programming.
Elliptic curve theory is presented via a simple, ground-up introduction to Curve25519 which
culminates in the Diffie‒Hellman process for establishing a shared secret in the presence of
an eavesdropper. Functional programming is presented by putting this theory into practice
via 100 lines of Erlang implementation followed by a script-based demonstration. The reader
will simultaneously develop a clearer view of elliptic curve cryptography theory that is also
connected to a functional Erlang implementation in practice.

A Tour of Curve25519 in Erlang
Introduction
This whitepaper provides an introduction to elliptic curve cryptography for casual users while delivering a brief tour
of the Erlang language for those interested in functional programming. Elliptic curve theory is presented via a simple,
ground-up introduction to Curve25519 which culminates in the Diffie‒Hellman process for establishing a shared secret
in the presence of an eavesdropper. Functional programming is presented by putting this theory into practice via 100
lines of Erlang implementation followed by a script-based demonstration. The reader will simultaneously develop
a clearer view of elliptic curve cryptography theory that is also connected to a functional Erlang implementation in
practice.
Elliptic curves are increasingly popular in transport layer security (TLS), signatures, certificates, messaging, blockchains 1
and most everything crypto.2 This is partly due to the potential for shorter keys, better performance, less power and
fewer side-channels relative to RSA and other alternatives. Unfortunately, even the well-informed person can find
the theory impenetrable, the implementation tricks obscure and the frequent assembly-language routines hard to
understand.
At the same time, the Erlang language and runtime system make some very different design choices relative to better
known alternatives such as JavaScript, Golang, Python, Rust and the C* family. Erlang is a functional programming
language that targets highly-concurrent, distributed, fault-tolerant and non-stop applications – in fact, the code can
be upgraded while the application continues to run. While it is instructive to consider such a different paradigm,
the typical “hello world” example followed by a review of language syntax and operators does not match everyone’s
learning style or available time-commitment.
This whitepaper has a strong focus on simplicity, clarity and brevity. The full source code file is available in Appendix:
Complete Erlang Code Listing on page 15 and the terminology is generally consistent with RFC 7748: Elliptic Curves
for Security 3 and RFC 8446: TLS 1.3.4 References are included to both provide additional context as well as support
deeper investigation into optimizations, corner cases and side-channel resistance. Note that the presented source
code is strictly educational, is not constant-time and it should not be used as-is in production.

Erlang
Let’s jump right into some Erlang. The biggest challenge to highly-concurrent systems 5 is the sharing of resources,
particularly data-structures in memory. Erlang essentially shares nothing and considers an application to be a collection of self-contained message-passing functions without side-effects or the concept of global variables. Further,
enigmatic race conditions can occur when reads and writes competing for the same resource happen in differing
orders. Largely for this reason, Erlang variables are strictly write-once,6 which is similar but even more extreme than
Rust’s concept of ownership and lifetimes. As this approach precludes loops, recursion is heavily utilized instead,
ideally using tail-recursion to minimize stack frame allocation. Process 7 creation is very inexpensive and the runtime
system comes with its own scheduler able to juggle ten thousand concurrent processes running on multiple cores
as the typical condition. Separate nodes can be further joined together into larger distributed systems 8 consisting of
workers, supervisors and application trees. Readers are encouraged to investigate the official Erlang documentation 9
for more insight into OTP components, generic behaviors and hot code-updating.

1

https://en.bitcoin.it/wiki/Secp256k1
https://ianix.com/pub/curve25519-deployment.html
3
https://tools.ietf.org/pdf/rfc7748.pdf
4
https://tools.ietf.org/html/rfc8446#page-96
5
https://en.wikipedia.org/wiki/Concurrent_computing
6
http://erlang.org/doc/reference_manual/expressions.html#variables
7
http://erlang.org/doc/reference_manual/processes.html
8
http://erlang.org/doc/reference_manual/distributed.html
9
https://erlang.org/doc/search/
2

File Prologue
Each Erlang file must have a few required metadata terms present, specifically the module name and function export
list. The module 10 name must match the file name and the functions visible to other modules must be explicitly
exported as shown on line 4 below. Erlang is very particular in that a precise function name includes its number of
parameters 11 (e.g. arity is the trailing /2 and /3 below for example). A list consists of comma-separated elements
enclosed in []. Finally, the language allows for additional and arbitrary metadata (such as -author and -spec) that
can be picked up by other tools besides the compiler.
In the code below, placing defined names in all caps is strictly by convention, the defined constant values are written
as base#value 12 integers where base is in the range 2 through 36, and functions/statements must end with a period.
Each of the specified constants will be described later in their particular usage context.
1

-module(curve25519).

2

-author("Eric Schorn").

3
4

-export([mul_k_u/2, test_k_u_iter/3]).

5
6

-define(PRIME,

16#7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFED).

7

-define(K_AND,

16#7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF8).

8

-define(K_OR,

16#4000000000000000000000000000000000000000000000000000000000000000).

9

-define(U_AND,

16#7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF).

10

-define(ALL256, 16#FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF).

11

-define(A24,

121665).

Modular Arithmetic
Cryptography is full of modular arithmetic 13 and this paper is no different. Think of operands and results as being
defined over a set of integers that wrap around. This is similar to a clock’s time that wraps around after 24 hours, a
video game’s score that wraps around after 100,000 points or even an American roulette wheel that wraps around
on its 38 individually-numbered pockets. The situation here is most like a novel roulette wheel where multiplicative
operations are used to wildly bounce a ball around – the ball being the integer coordinates of a point on a curve.
A clock, game or roulette wheel that wraps around on an even integer presents an issue. If our bouncing ball starts on
an even integer and that integer is multiplied by any other integer, the result can only be another even integer, making
the odd integers unreachable. In fact, the same issue applies to schemes that wrap around any number with integer
factors; the factors can cause certain destinations to be unreachable by our multiplicative operations. For this reason,
a prime modulus 14 (e.g. a wrap-around number without any factors) is often used to make sure all the possibilities
remain in play. A ring 15 is loosely defined as this set of possible numbers along with two binary operators that generate
them. Some curves are also defined over a prime power pn which p is a prime and n > 1. For Curve25519, such a huge
prime modulus is used that its size roughly approaches the number of atoms in the observable universe. This is the
first constant PRIME defined on line 6 above and used as ?PRIME on lines 16, 21 and 26 in the subsequent code below.
Below are three basic functions in Erlang that support modular arithmetic. They each perform the intuitive arithmetic
operation followed by a clause that drives the wrap-around behavior. The function name and parameters are first
10

http://erlang.org/doc/reference_manual/modules.html
http://erlang.org/doc/reference_manual/functions.html
12
http://erlang.org/doc/reference_manual/data_types.html
13
https://en.wikipedia.org/wiki/Modular_arithmetic
14
https://crypto.stanford.edu/pbc/notes/numbertheory/gen.html
15
http://mathworld.wolfram.com/Ring.html

declared and the -> separator then precedes the implementation, which ends with a period. In each case shown,
the implementation consists of a single expression whose value is returned. Comments are required to begin with %,
though a more common style starts with %%. The rem operator 16 used in each function denotes the remainder after
integer division. Variable names must start in upper case.
14

-spec(add(X :: non_neg_integer(), Y :: non_neg_integer()) -> non_neg_integer()).

15

add(X, Y) ->
(X + Y) rem ?PRIME.

16
17
18
19

-spec(sub(X :: non_neg_integer(), Y :: non_neg_integer()) -> non_neg_integer()).

20

sub(X, Y) ->

%% Y - Y = 0 = PRIME

so

-Y = PRIME - Y

(X + ?PRIME - Y) rem ?PRIME.

21
22
23
24

-spec(mul(X :: non_neg_integer(), Y :: non_neg_integer()) -> non_neg_integer()).

25

mul(X, Y) ->
(X * Y) rem ?PRIME.

26

Erlang is a dynamically typed language, so the -spec 17 metadata directives seen preceding each function declaration
are for an external type checker called Dialyzer.18 Consider these lines as comments that formally declare a function’s
input and output types. Strict type checking is a separate development step and irrelevant at runtime (unless things
fail of course).
Again, the three functions above look like ordinary arithmetic with the exception of the rem ?PRIME clause. This clause
divides an interim result by the prime modulus and returns the remainder to enforce the wrap-around behavior. In
modular arithmetic, the modulus ?PRIME is considered equivalent (or congruent) to zero,19 and this fact helps recast
the subtraction logic to always return an unsigned positive result. Since Erlang has arbitrary precision integers, there
is no need to worry about overflow. Also, as indicated by -spec, all function parameters in this code are intended to
be non-negative integers as are all intermediate values.
It is clear at first glance that the addition and subtraction functions are related in the sense that they reverse each other.
Any number can be added to an initial value, that number subsequently subtracted from the sum, and the initial value
reappears. This works even in the presence of the ?PRIME modulus wrapping behavior. For all values X there is an
additive inverse value Y, such that X - Y = 0. As evident in the sub() code above, -Y is calculated as simply PRIME Y and is thus always a non-negative integer. However, things are not as simple for the multiplication function’s reverse
sibling function. Here, a function is needed that returns the multiplicative inverse such that a value times its inverse
equals one. A multiplication table mod 19 is shown below, where the entries containing 1 are significant to this goal.
The prime modulus 19 was chosen as an example that nicely fits the printed page. Who would have guessed that the
square root of 6 mod 19 is both 5 and 14 (since 5*5 and 14*14 mod 19 both equal 6)?!
16

http://erlang.org/doc/reference_manual/expressions.html#arithmetic-expressions
http://erlang.org/doc/reference_manual/typespec.html#specifications-for-functions
18
http://erlang.org/doc/apps/dialyzer/dialyzer_chapter.html
19
https://www.khanacademy.org/computing/computer-science/cryptography/modarithmetic/a/congruence-modulo
17

Figure 1: Multiplication Mod 19

First, a note on reading the table. Consider any particular entry: the leftmost row header in blue represents a multiplier
operand while the topmost column header in blue represents the multiplicand operand, and the intersecting entry
contains the multiplied result mod 19. The most interesting entries are 1 because that identifies two operands
(leftmost and topmost) that are the inverse of each other – when multiplied their result is 1. There is no simple and
straightforward pattern to the location of 1’s and thus no simple and computationally cheap way of finding a value’s
inverse. A little complexity enters here based on Fermat’s little theorem 20 which states that (when p is a prime):

xp ≡ x mod p
So if each side is multiplied by x−2 the very useful result is:

xp−2 ≡ x−1

mod p

Note that the congruence 21 relation is used above (rather than strict equality), which means each side has the identical
remainder after being divided by the modulus. This is the key to our operating definition of a reverse sibling to the
multiplication function. If an operand X is exponentiated to the power of PRIME-2 as on the left side of the equation,
the result is the multiplicative inverse of X as on the right side of the equation.
Exponentiation is just repeated multiplication (in the same way that multiplication is just repeated addition, a useful
fact used later on). However, the value of the exponent (PRIME-2) presents an issue because its large size precludes
a straightforward loop of one-after-another multiplications. Iterating through one multiplication at a time would take
(literally) forever. However, consider the following facts:

x2 = x · x and x4 = (x2 )2 and x8 = ((x2 )2 )2 and x16 = (((x2 )2 )2 )2 ) and x31 = x · x2 · x4 · x8 · x16
20
21

http://mathworld.wolfram.com/FermatsLittleTheorem.html
https://en.wikipedia.org/wiki/Modular_arithmetic#Congruence_classes

256

So, a continued process of squaring operand X can deliver x2
in only 255 operations as well as generate all the
interim powers of two in the process. Consider a routine that repeatedly squares the operand X while inspecting each
successive bit of the exponent value from the rightmost LSB to the leftmost MSB. When the individual exponent bit of
interest is one, the routine multiplies the current square into an interim result. The end result after iterating through
the entire exponent is then the inverse of operand X. Here is the Erlang code for the inverse function. Note that it is
using our custom mul() function that takes the remainder after each step to prevent the operand’s magnitude from
exploding.
29
30
31
32

-spec(inv(X :: non_neg_integer(), Exp :: non_neg_integer(), Result :: non_neg_integer())
-> non_neg_integer()).
inv(_X, 0, Result) ->
Result;

33
34
35
36

inv(X, Exp, Result) ->
if
Exp band 1 == 1->

40

R1 = mul(Result, X);
true ->
R1 = Result
end,

41

X1 = mul(X, X),

42

inv(X1, Exp bsr 1, R1).

37
38
39

In Erlang, it is not unusual to see multiple versions of the same function. When called, the system will attempt to match
the function name, arity and operands with each declaration in order and then execute the first match. If the above
collection of inv() functions were invoked with the middle operand set to 0, the topmost declaration on line 31 would
match and execute. Otherwise, the second declaration on line 34 matches and executes. Note how expressions are
separated by a comma, function clauses are separated by a semicolon and both function declaration/bodies together
form a single unit ending with a period. The if statement starting on line 35 is similarly a series of clauses that are
attempted to match in order, which is why the final true value represents the traditional else clause, as it matches
everything left over. The _X parameter in the first inv() function declaration on line 31 tells the compiler that this
placeholder parameter will not be used.
The lack of looping in Erlang forces a recursive approach. The second portion of the inv() function above multiplies
the operand into a working result if the exponent LSB=1 (lines 36-37), repeatedly squaring the operand (line 41), and
then recursively calls itself with an exponent right shifted by 1 on line 42. The binary shift right operator is denoted
bsr which helps step through the exponent bits, while band is the binary-and operator which helps inspect the LSB
value. When the repeatedly-shifted exponent reaches zero, the top function declaration is matched on line 31 and the
result finally returned. Line 42 in the second function is an example of tail-call recursion where no further calculation
is done so as to obviate the need for a fresh stack frame to be allocated on each iteration.
The above function(s) could easily be tested by taking random values, multiplying each by the calculated inverse value
and checking for a result of 1. The multiply and inverse functions are a great example of how some reversible functions
can be far easier to run in one direction relative to the reverse direction. That difference will get even more extreme
shortly.

Curve25519
General elliptic curves are specified in a number of forms including:
• The short Weierstrass equation y 2 = x3 + ax + b where 4a3 + 27b2 is nonzero.

• The Edwards equation x2 + y 2 = 1 + dx2 y 2 where d(1 − d) is nonzero.
• The Montgomery equation By 2 = x3 + Ax2 + x where B(A2 − 4) is nonzero.
The forms above are just a jumping-off point to a tremendous amount of variety and complexity. The nonzero requirements in each form prevents subsequent calculations from reaching non-singluar scenarios. NIST has specified
a large collection of curves and parameters in the draft NIST SP 800-186 Recommendations for Discrete LogarithmBased Cryptography: Elliptic Curve Domain Parameters,22 including Curve25519. As an aside, Bitcoin uses a curve
called secp256k1.23 This diversity is also a source of significant controversy within the cryptographic community
related to choices of curve forms and constants.24 One result of these ‘arguments’ is the rapidly increasing adoption
of Curve25519.
Curve25519 is a specific Montgomery equation with certain A and B constants, and defined over a particular field of
integers. When a point on the curve is referenced, its coordinates represent a solution to the curve equation. For
consistency, the coordinates will be labelled u and v as used in RFC 7748. The precise Curve25519 equation written in
its Montgomery form is:

v 2 = u3 + 486662 · u2 + u

mod 2255 − 19

Because curve points can be added, multiplied by a scalar and a few other properties hold, they form a group.25 The
number of discrete points on our ‘novel roulette wheel’ (e.g. elliptic curve), which is known as the curve order, is 8 · l
where l is 2252 + 27742317777372353535851937790883648493. Setting aside integers, modular arithmetic and the
prime modulus for a moment, the figure below depicts a portion of the curve plotted over real numbers:

Figure 2: Elliptic Curve Over Real Numbers

Now let’s define an operation called point addition that really just moves from one pair of points (or curve solutions) to
another point (or curve solution). Even for the case of real numbers, if the initial points are rational numbers, then the
22

https://nvlpubs.nist.gov/nistpubs/SpecialPublications/NIST.SP.800-186-draft.pdf
https://en.bitcoin.it/wiki/Secp256k1
24
https://bada55.cr.yp.to/bada55-20150927.pdf
25
https://en.wikipedia.org/wiki/Group_theory
23

result will be rational as well. The same is true for our ring of integers. Thus, these calculations will apply to our actual
code despite our working with real numbers for the moment. For simplicity, the ‘point at infinity’ will be ignored.26
Here is how point addition works. Start with two points P = (u1 , v1 ) and Q = (u2 , v2 ). Draw a line between them and
derive the equation for that line. That line will intersect the curve in a third point called R’ at (u3 , -v3 ). Now, the original
curve equation, the derived line equation and the two original point solutions are sufficient to solve for the coordinates
of R’ as the third solution. The final point R sum results from just negating the calculated v3 coordinate of R’, which
effectively flips it across the X axis. A graphical representation of this procedure is shown below.

Figure 3: Point Addition on an Elliptic Curve

The solution to the coordinates of R in equation form is:

u3 =

y3 =

(u2 v1 − u1 v2 )2
u1 u2 (u2 − u1 )2

(2u1 + u2 + 486662)(v2 − v1 )
(v2 − v1 )3
−
− v1
u2 − u1
(u2 − u1 )3

If points P and Q were the same point, then the intersecting line equation would be tangent to the curve at point P
and this is now called point doubling with slightly more elaborate math driving different equation results shown below.
The general idea works exactly the same as above.
For the point doubling case, we get R(u3 , v3 ) as:

u3 =

v3 =
26

(u21 − 1)2
4u1 (u21 + 486662u1 + 1)

(2u1 + u1 + 486662)(3u21 + 2 · 486662u1 + 1) (3u21 + 2 · 486662u1 + 1)3
−
− v1
2v1
(2v1 )3

https://math.stackexchange.com/questions/1118838/elliptic-curve-point-at-infinity

While the above result looks more complicated rather than less complicated, note how u can be fully calculated without
v . The doubling calculation can be repeated without involving v , and the v value later resolved by the original curve
equation if/when needed, meaning we can operate on a point by using only its u coordinate.
Recall how integer exponentiation was just repeated integer multiplication. Now, point multiplication is just repeated
point addition. To be very clear, in this case we are multiplying a scalar times a point, not one point times another
point. The same coordinate squaring strategy for exponentiation translates into point doubling strategy for point
multiplication. So the same style of calculation can be run - iterate over each bit of the multiplier operand and
add a repeatedly-doubled point into the temporary result as appropriate. The multiplier is a scalar called K and the
multiplicand is a point called U (with the coordinate u). The above procedure is a brief description of the Montgomery
ladder 27 algorithm as implemented below.
50

-spec(mul_k_u(K :: non_neg_integer(), U :: non_neg_integer()) -> non_neg_integer()).

51

mul_k_u(K, U) ->

%% x_1 = u, x_2 = 1, z_2 = 0, x_3 = u, z_3 = 1, swap = 0

52

K1 = binary:decode_unsigned(<<K:256/little-unsigned-integer-unit:1>>),

53

K2 = (K1 band ?K_AND) bor ?K_OR,

54
55

U1 = binary:decode_unsigned(<<U:256/little-unsigned-integer-unit:1>>),
U2 = U1 band ?U_AND,

56

mul_k_u(254, K2, U2, 1, 0, U2, 1, 0).

57
58
59

-spec(mul_k_u(T :: -1..254, _K :: non_neg_integer(), _X_1 :: non_neg_integer(), X_2 ::

60

non_neg_integer(), Z_2 :: non_neg_integer(), X_3 ::non_neg_integer(), Z_3 ::

61

non_neg_integer(), Swap :: 0..1) -> non_neg_integer()).

62

mul_k_u(T, _K, _X_1, X_2, Z_2, X_3, Z_3, Swap) when T == -1 ->
{X_2a, _X_3a} = cswap(Swap, X_2, X_3),

%% (x_2, x_3) = cswap(swap, x_2, x_3)

64

{Z_2a, _Z_3a} = cswap(Swap, Z_2, Z_3),

%% (z_2, z_3) = cswap(swap, z_2, z_3)

65

Inverse = inv(Z_2a),

%% Return x_2 * (z_2^(p - 2))

66

Result = mul(X_2a, Inverse),

67

binary:decode_unsigned(<<Result:256/little-unsigned-integer-unit:1>>);

63

68
69

mul_k_u(T, K, X_1, X_2, Z_2, X_3, Z_3, Swap) ->

%% For t = bits-1 down to 0:

71

K_t = (K bsr T) band 1,
Swap_a = Swap bxor K_t,

%%
%%

k_t = (k >> t) & 1
swap ^= k_t

72

{X_2a, X_3a} = cswap(Swap_a, X_2, X_3),

%%

(x_2, x_3) = cswap(swap, x_2, x_3)

73
74

{Z_2a, Z_3a} = cswap(Swap_a, Z_2, Z_3),
Swap_b = K_t,

%%
%%

(z_2, z_3) = cswap(swap, z_2, z_3)
swap = k_t

75

A = add(X_2a, Z_2a),

%%

A = x_2 + z_2

76

AA = mul(A, A),

%%

AA = A^2

77

B = sub(X_2a, Z_2a),

%%

B = x_2 - z_2

78

BB = mul(B, B),

%%

BB = B^2

79

E = sub(AA, BB),

%%

E = AA - BB

80

C = add(X_3a, Z_3a),

%%

C = x_3 + z_3

81

D = sub(X_3a, Z_3a),

%%

D = x_3 - z_3

82

DA = mul(D, A),

%%

DA = D * A

83

CB = mul(C, B),

%%

CB = C * B

70

27

https://eprint.iacr.org/2017/293.pdf

84

XX1 = add(DA, CB),

85

X_3b = mul(XX1, XX1),

%%

x_3 = (DA + CB)^2

86

XX2 = sub(DA, CB),

%%

z_3 = x_1 * (DA - CB)^2

87

XX3 = mul(XX2, XX2),

88

Z_3b = mul(X_1, XX3),

89
90

X_2b = mul(AA,BB),

%%

x_2 = AA * BB

XX4 = mul(?A24, E),

%%

z_2 = E * (AA + a24 * E)

91

XX5 = add(AA, XX4),

92

Z_2b = mul(E, XX5),

93

mul_k_u(T - 1, K, X_1, X_2b, Z_2b, X_3b, Z_3b, Swap_b).
As mentioned early on, a specific function name includes its arity. So the first function starting on line 51 above is
mul_k_u/2 and is exported as described at the start. This is a different function than the two that follow with the same
name starting on lines 62 and 69, which are both internal private functions. Take note of the various punctuations
involved in the Erlang syntax which was described earlier.
Let’s start with the very bottom function starting on line 69. It is intended to iterate from T=254 down to 0 (inclusive)
recursively, where the original T is specified elsewhere. Note the tighter -spec for T on line 59. This function iterates
through each individual bit of the K operand from left MSB to right LSB, and performs a point doubling and optionally
point addition. The exact logic derivation will be skipped as the code is conveniently adapted from the pseudo-code
in RFC 7748 28 which remains in the comments on the right for reference. Note that variables are only assigned once
which requires a little adaptation but really doesn’t present much trouble at all. While the code would clearly benefit
from an optimized squaring function, it isn’t strictly necessary as mul(x, x) will suffice, and thus skipped for brevity.
The function at the top starting on line 51 launches the calculation. It decodes K in the proper endian format then
performs AND and OR with two predefined constants to mask and set particular bits as directed by RFC 7748. The
point U also takes a little massaging with one predefined constant to mask the most significant bit of the u coordinate.
Then the bottom function is called on line 56 where T is set to 254.
When the main recursive (bottom, lines 69-93) function’s iterative index T goes below 0, the middle function on line
62 is matched and executed, which does a little final calculation (including the intriguing inverse operation described
earlier) before ultimately returning the final calculated result.
Recall that the above calculations are all performed on discrete integers in practice, rather than the real numbers
shown in the smooth and intuitive curves earlier. Shown below is the same curve equation plotted over integers mod
19.29 As mentioned earlier, starting with a discrete point solution and following the process elaborated above will result
in another discrete solution. However, the geometric analogy becomes very difficult to see in the discrete space.
28
29

https://tools.ietf.org/html/rfc7748#section-5
http://www.graui.de/code/ffplot/

Figure 4: Elliptic Curve Mod 19

In the end, consider the point solution to be like a ball bouncing around a huge roulette wheel in a crazy out-of-order
and difficult to reverse fashion. The size of the wheel roughly approaches to the number of observable atoms in the
universe. The number of wheel spins also roughly approaches to the number of observable atoms in the universe.
The resulting point solution pattern is believed to be impossible to work backwards. The (believed to be) intractable
task of reversing the point multiplication is known as the elliptic curve discrete logarithm problem 30, .31
The fastest known algorithm to solve the elliptic curve discrete logarithm problem runs in time proportional to the
square root of the field size. For Curve25519, this size is above 2252 thus yielding approximately 128-bits of security.
For comparison, NIST suggests a 3072-bit key for comparable security involving RSA algorithms[ˆABC].

Cswap
The sharp-eyed reader will spot a single function cswap() used earlier that hasn’t been defined yet. Here it is for
completeness.
96

-spec(cswap(Swap :: 0..1, X2 :: non_neg_integer(), X3 :: non_neg_integer()) ->
{non_neg_integer(), non_neg_integer()}).

97
98

cswap(Swap, X2, X3) ->

101

Dummy = Swap*?ALL256 band (X2 bxor X3),
X2a = X2 bxor Dummy,
X3a = X3 bxor Dummy,

102

{X2a, X3a}.

99
100

The above function is a conditional swap, and explains the need for the ALL256 constant used on line 99. If the control
30
31

https://eprint.iacr.org/2015/1022.pdf
https://andrea.corbellini.name/2015/05/23/elliptic-curve-cryptography-finite-fields-and-discrete-logarithms/

input is 1, then the operands are swapped, otherwise not.

Test vectors
Finally, RFC 7748 has a ‘smoke test’ where the initial inputs are set to a particular value and the function repeatedly
run with outputs fed back into the inputs. The idea is that a million circular iterations with a specific starting point and
known ending point lends confidence to the correctness of intermediate calculations. Here is our variant of this with
some relevant comments included in the source code.
105
106
107
108
109
110
111
112

-spec(test_k_u_iter(K :: non_neg_integer(), U :: non_neg_integer(), Iter ::
non_neg_integer()) -> ok).
test_k_u_iter(K, U, Iter) when Iter > 0 ->
%% For each iteration, set k to be the result of calling the function and
%% u to be the old value of k. The final result is the value left in k.
K1 = mul_k_u(K, U),
U1 = K,
test_k_u_iter(K1, U1, Iter - 1);

113
114
115
116
117
118
119
120

test_k_u_iter(K, _U, 0) ->
io:fwrite("Result is ~.16B~n", [K]).
%% curve25519:test_k_u_iter(
%%
16#0900000000000000000000000000000000000000000000000000000000000000,
%%
16#0900000000000000000000000000000000000000000000000000000000000000, 1000).
%% After 1K iterations: 684cf59ba83309552800ef566f2f4d3c1c3887c49360e3875f2eb94d99532c51
%% After 1M iterations: 7c3911e0ab2586fd864497297e575e6f3bc601c0883c30df5f4dd2d24f665424
The top function on line 107 starts the process and recursively iterates back on itself until the index hits 0. Note some
new syntax on line 107 involving when Iter > 0 which prevents the function from executing on the last iteration and
allows the process to fall through to the next declaration. At that point the bottom function on line 114 prints out the
final result. A million iterations takes a little time, but delivers the correct overall result. Note that this does not address
all possible inputs or side-channels.

The Elliptic Curve Diffie-Hellman (ECDH) Scenario
TLS 1.3 specifies a protocol to establish a shared secret that uses Curve25519 as described above. It effectively specifies
the curve, its constants and an initial starting point G called a base point. Note that point multiplication is commutative.
Alice and Bob want to establish a private conversation in the presence of Eve the eavesdropper. So, Alice chooses a
large random number Ka to serve as her private key. She multiplies the agreed base point G by her private key and
sends the resulting public key Ka ∗ G to Bob in plain text. Similarly, Bob chooses a large random number Kb to serve
as his private key. He multiplies the same agreed base point G by his private key and sends the resulting public key
Kb ∗ G to Alice in plain text. The private keys are never disclosed to anyone else; only the public keys resulting from
the scalar-times-point multiplication are sent in the clear.
Now Alice can take her received Kb ∗ G point from Bob and multiply it again by her private Ka key. Similarly, Bob can
take his received Ka ∗ G point from Alice and again multiply it by his private Kb . In this way, they both arrive at the
common result Ka ∗ Kb ∗ G. This is their shared secret. While Eve has both public keys Ka ∗ G and Kb ∗ G, recall that
the elaborate calculation described above is scalar times point multiplication (not point times point), so the pair Ka ∗ G
and Kb ∗ G is not usable. There is believed to be no simple way for Eve to factor out the G multiplicand in either public
key except through brute force. Thus, Bob and Alice can now use Ka ∗ Kb ∗ G as the basis for a symmetric key and

continue their conversation in private despite the presence of Eve.

Escript
Erlang also provides a scripting environment called escript, which is a fast and simple way of testing out the scenario
described above. The script is put in a normal text file and can either be run with the escript command or marked as
an executable and executed directly via the correct shebang 32 (exactly as one might do with Python). The Erlang
scripting environment expects to invoke the main/1 function by default (shown below on line 3) with arguments
optionally supplied from the command line. In our case the _ placeholder is used to indicate that the arguments
are not used. Sometimes a placeholder name is post-pended to serve as a human reminder – this occurred in the first
declaration of inv/2 on line 31 and the first declaration of mul_k_u/8 on line 62 above. In any event, the full script
shown below runs the RFC 7748 ‘smoke test’ followed by the scenario described above.
1

#!/usr/bin/escript

2
3

main(_) ->

4

6

%% Run 1M testcase
io:format("Testing...~n"),

7

io:format("Expecting 7c3911e0ab2586fd864497297e575e6f3bc601c0883c30df5f4dd2d24f665424~n"),

8

curve25519:test_k_u_iter(
16#0900000000000000000000000000000000000000000000000000000000000000,
16#0900000000000000000000000000000000000000000000000000000000000000,

5

9
10

1000000),

11
12
13

%% Force strong randomness and declare common point G

14

_ = crypto:rand_seed(),
G = 9,

15
16
17

%% Alice and Bob generate private keys, never shared directly

18

Alice_private_key = rand:uniform(16#7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF),

19

Bob_private_key = rand:uniform(16#7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF),

20
21

%% Alice and Bob generate public keys to share with each other in the presence of Eve

22

Alice_public_key = curve25519:mul_k_u(Alice_private_key, G),

23

Bob_public_key = curve25519:mul_k_u(Bob_private_key, G),

24
25

%% Alice and Bob use their private key and shared key to separately derive a shared secret

26

Alice_shared_key = curve25519:mul_k_u(Alice_private_key, Bob_public_key),

27

Bob_shared_key = curve25519:mul_k_u(Bob_private_key, Alice_public_key),

28
29

%% Alice encrypts a secret message with their shared key and sends it to Bob with an IV

30

IV = crypto:strong_rand_bytes(16),

31

CipherText = crypto:block_encrypt(aes_cbc, IV,
<<Alice_shared_key:128/little-unsigned-integer-unit:1>>, <<"A secret message">>),

32
33
34

%% Bob decrypts the secret message with their shared key and prints to terminal

35

PlainText = crypto:block_decrypt(aes_cbc, IV,
<<Bob_shared_key:128/little-unsigned-integer-unit:1>>, CipherText),

36
32

https://en.wikipedia.org/wiki/Shebang_(Unix)

io:format("~s~n", [PlainText]).

37

%% <---- Sucess?

The Erlang script runs exactly as expected. After successfully running the ‘smoke test’, Alice and Bob create their
individual private keys, calculate and share the corresponding public keys in the presence of Eve, and finally calculate
a shared secret used to (symmetrically) encrypt and decrypt a secret message.

Figure 5: Secret message encrypted then decrypted under a Diffie-Hellman shared secret

It can be seen that the private keys are kept private and Eve’s knowledge of the shared public keys is not sufficient to
determine Alice’s and Bob’s shared secret. Thus, Alice and Bob are able to communicate privately despite the presence
of Eve. For clarity, this scenario is not intended to address man-in-the-middle situations which are mitigated by other
means; the reader is referred to RFC 8446 33 describing TLS 1.3 for this larger context.

Conclusion
This whitepaper has introduced elliptical curve cryptography from the ground up in the context of a functional Erlang
implementation. All code has been presented and the novel Erlang syntax described. A script demonstrates functionality and the Diffie-Hellman process to establish a shared secret in the presence of an eavesdropper. The reader
is now well placed to further investigate both elliptic curve theory and practice, including topics such as constanttime requirements, input validation conditions, alternative algorithm/implementation approaches, optimizations of
both point arithmetic and ‘big number’ field arithmetic, as well as alternative curve equations and their associated
arithmetic laws. The reader is further positioned to study additional functional programming concepts in Erlang.

33

https://tools.ietf.org/html/rfc8446

Appendix: Complete Erlang Code Listing
The following code is a simple and functional implementation of Curve25519 in Erlang. As this is meant to be a strictly
educational example, simplicity is favored over robustness, constant-time characteristics, and checked corner cases.
1

-module(curve25519).

2

-author("Eric Schorn").

3
4

-export([mul_k_u/2, test_k_u_iter/3]).

5
6

-define(PRIME,

16#7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFED).

7

-define(K_AND,

16#7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF8).

8

-define(K_OR,

16#4000000000000000000000000000000000000000000000000000000000000000).

-define(U_AND,

16#7FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF).

9
10

-define(ALL256, 16#FFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFFF).

11

-define(A24, 121665).

12
13
14

-spec(add(X :: non_neg_integer(), Y :: non_neg_integer()) -> non_neg_integer()).

15

add(X, Y) ->

16

(X + Y) rem ?PRIME.

17
18
19

-spec(sub(X :: non_neg_integer(), Y :: non_neg_integer()) -> non_neg_integer()).

20

sub(X, Y) ->

21

%% Y - Y = 0 = PRIME

so

-Y = PRIME - Y

(X + ?PRIME - Y) rem ?PRIME.

22
23
24

-spec(mul(X :: non_neg_integer(), Y :: non_neg_integer()) -> non_neg_integer()).

25

mul(X, Y) ->

26

(X * Y) rem ?PRIME.

27
28
29

-spec(inv(X :: non_neg_integer()) -> non_neg_integer()).

30

inv(X) ->

31

inv(X, ?PRIME - 2, 1).

32
33
34
35
36
37

-spec(inv(X :: non_neg_integer(), Exp :: non_neg_integer(), Result :: non_neg_integer())
-> non_neg_integer()).
inv(_X, 0, Result) ->
Result;

38
39
40
41

inv(X, Exp, Result) ->
if
Exp band 1 == 1->

45

R1 = mul(Result, X);
true ->
R1 = Result
end,

46

X1 = mul(X, X),

47

inv(X1, Exp bsr 1, R1).

42
43
44

48
49
50

-spec(mul_k_u(K :: non_neg_integer(), U :: non_neg_integer()) -> non_neg_integer()).

51

mul_k_u(K, U) ->

%% x_1 = u, x_2 = 1, z_2 = 0, x_3 = u, z_3 = 1, swap = 0

52

K1 = binary:decode_unsigned(<<K:256/little-unsigned-integer-unit:1>>),

53

K2 = (K1 band ?K_AND) bor ?K_OR,

54
55

U1 = binary:decode_unsigned(<<U:256/little-unsigned-integer-unit:1>>),
U2 = U1 band ?U_AND,

56

mul_k_u(254, K2, U2, 1, 0, U2, 1, 0).

57
58
59
60
61
62

-spec(mul_k_u(T :: -1..254, _K :: non_neg_integer(), _X_1 :: non_neg_integer(), X_2 ::
non_neg_integer(), Z_2 :: non_neg_integer(), X_3 ::non_neg_integer(), Z_3 ::
non_neg_integer(), Swap :: 0..1) -> non_neg_integer()).
mul_k_u(T, _K, _X_1, X_2, Z_2, X_3, Z_3, Swap) when T == -1 ->

63

{X_2a, _X_3a} = cswap(Swap, X_2, X_3),

%% (x_2, x_3) = cswap(swap, x_2, x_3)

64

{Z_2a, _Z_3a} = cswap(Swap, Z_2, Z_3),

%% (z_2, z_3) = cswap(swap, z_2, z_3)

65

Inverse = inv(Z_2a),

%% Return x_2 * (z_2^(p - 2))

66

Result = mul(X_2a, Inverse),

67

binary:decode_unsigned(<<Result:256/little-unsigned-integer-unit:1>>);

68
69

mul_k_u(T, K, X_1, X_2, Z_2, X_3, Z_3, Swap) ->

%% For t = bits-1 down to 0:

71

K_t = (K bsr T) band 1,
Swap_a = Swap bxor K_t,

%%
%%

k_t = (k >> t) & 1
swap ^= k_t

72

{X_2a, X_3a} = cswap(Swap_a, X_2, X_3),

%%

(x_2, x_3) = cswap(swap, x_2, x_3)

73
74

{Z_2a, Z_3a} = cswap(Swap_a, Z_2, Z_3),
Swap_b = K_t,

%%
%%

(z_2, z_3) = cswap(swap, z_2, z_3)
swap = k_t

75

A = add(X_2a, Z_2a),

%%

A = x_2 + z_2

76

AA = mul(A, A),

%%

AA = A^2

77

B = sub(X_2a, Z_2a),

%%

B = x_2 - z_2

78

BB = mul(B, B),

%%

BB = B^2

79

E = sub(AA, BB),

%%

E = AA - BB

80

C = add(X_3a, Z_3a),

%%

C = x_3 + z_3

81

D = sub(X_3a, Z_3a),

%%

D = x_3 - z_3

82

DA = mul(D, A),

%%

DA = D * A

83

CB = mul(C, B),

%%

CB = C * B

84

XX1 = add(DA, CB),

%%

x_3 = (DA + CB)^2

85

X_3b = mul(XX1, XX1),

86

XX2 = sub(DA, CB),

%%

z_3 = x_1 * (DA - CB)^2

87

XX3 = mul(XX2, XX2),

88

Z_3b = mul(X_1, XX3),

89

X_2b = mul(AA,BB),

%%

x_2 = AA * BB

90

XX4 = mul(?A24, E),

%%

z_2 = E * (AA + a24 * E)

91

XX5 = add(AA, XX4),

92

Z_2b = mul(E, XX5),

93

mul_k_u(T - 1, K, X_1, X_2b, Z_2b, X_3b, Z_3b, Swap_b).

70

94

95
96
97
98

-spec(cswap(Swap :: 0..1, X2 :: non_neg_integer(), X3 :: non_neg_integer()) ->
{non_neg_integer(), non_neg_integer()}).
cswap(Swap, X2, X3) ->

101

Dummy = Swap*?ALL256 band (X2 bxor X3),
X2a = X2 bxor Dummy,
X3a = X3 bxor Dummy,

102

{X2a, X3a}.

99
100

103
104
105
106
107
108
109
110
111
112

-spec(test_k_u_iter(K :: non_neg_integer(), U :: non_neg_integer(), Iter ::
non_neg_integer()) -> ok).
test_k_u_iter(K, U, Iter) when Iter > 0 ->
%% For each iteration, set k to be the result of calling the function and
%% u to be the old value of k. The final result is the value left in k.
K1 = mul_k_u(K, U),
U1 = K,
test_k_u_iter(K1, U1, Iter - 1);

113
114
115
116
117
118
119
120

test_k_u_iter(K, _U, 0) ->
io:fwrite("Result is ~.16B~n", [K]).
%% curve25519:test_k_u_iter(
%%
16#0900000000000000000000000000000000000000000000000000000000000000,
%%
16#0900000000000000000000000000000000000000000000000000000000000000, 1000).
%% After 1K iterations: 684cf59ba83309552800ef566f2f4d3c1c3887c49360e3875f2eb94d99532c51
%% After 1M iterations: 7c3911e0ab2586fd864497297e575e6f3bc601c0883c30df5f4dd2d24f665424

